

---

---

# The Essence of Reactive Programming

*A Theoretical Approach*

---

---

By

EDDY BERTOLUZZO



Faculty of Electrical Engineering, Mathematics and Computer Science  
DELFT UNIVERSITY OF TECHNOLOGY

A dissertation submitted to Delft University of Technology in  
accordance with the requirements of the MASTER OF SCIENCE  
degree in the Faculty of Electrical Engineering, Mathematics and  
Computer Science.

OCTOBER 2016



## ABSTRACT

In the last few years the \*\*\* of Reactive Programming has gained much traction in the developer's community, especially with the general trend for programming languages to embrace functional programming and the shifting of today's applications towards an asynchronous approach to modeling data. The goal of this work is to shed some light to the sea of more or less informed opinions regarding what the meaning of Reactive Programming is thorough the use of Mathematics...

Finish it!



## DEDICATION AND ACKNOWLEDGEMENTS

*"Well, here at last, dear friends,  
on the shores of the Sea comes the  
end of our fellowship in  
Middle-earth. Go in peace! I will not  
say: do not weep, for not all tears are  
an evil."*

---

— J.R.R. Tolkien,  
*The Return of the King*

H ere goes the dedication.



## TABLE OF CONTENTS

	Page
<b>Introduction</b>	<b>1</b>
Contributions . . . . .	2
Overview . . . . .	3
Notation & Conventions . . . . .	3
<b>1 Reactive Programming</b>	<b>5</b>
1.1 The Essence of Reactive Programs . . . . .	5
1.2 Why Reactive Programming Matters . . . . .	6
1.3 Related Work . . . . .	7
1.3.1 Reactive Extensions . . . . .	7
1.3.2 Reactive Streams . . . . .	7
1.3.3 Functional Reactive Programming . . . . .	7
<b>2 Into the Rabbit Hole: <i>Deriving the Observable</i></b>	<b>9</b>
2.1 Iterables . . . . .	10
2.2 The Essence of Iterables . . . . .	11
2.3 Applying Duality . . . . .	17
2.4 Termination and Error Handling . . . . .	23
2.5 Formalizing Observables . . . . .	24
<b>3 Out of the rabbit hole: <i>Towards a usable API</i></b>	<b>29</b>
3.1 The Reactive Contract . . . . .	30
3.1.1 Reactive Grammar . . . . .	30
3.1.2 Serialized Messages . . . . .	31
3.1.3 Best Effort Cancellation . . . . .	31
3.1.4 Resource Cleanup After Termination . . . . .	31
3.1.5 A note on the Reactive Contract . . . . .	32
3.2 Concurrency with Schedulers . . . . .	32
3.2.1 A Note on the Concept of Time . . . . .	35
3.2.2 A Note on Orthogonality . . . . .	35

## TABLE OF CONTENTS

---

3.3	Subscriptions . . . . .	35
3.3.1	Impact on Schedulers . . . . .	39
3.3.2	Formalizing Subscriptions . . . . .	41
3.4	Operators . . . . .	42
3.4.1	Functor, Applicative, Monad . . . . .	43
3.4.2	The Essential Operators . . . . .	47
3.4.3	Formalizing Operators . . . . .	47
	<b>Conclusion</b>	<b>51</b>
	Future Work . . . . .	51
	Conclusion . . . . .	51
<b>A</b>	<b>Appendix A</b>	<b>55</b>
A.1	Categorical Duality . . . . .	55
A.2	Continuations . . . . .	55
A.3	(Co)Products . . . . .	55
A.4	(Un)Currying . . . . .	55
A.5	(Covariance & Contravariance . . . . .	55
A.6	Functors . . . . .	55
<b>B</b>	<b>Appendix B</b>	<b>57</b>
	<b>Bibliography</b>	<b>59</b>



## INTRODUCTION

*"Lasciate ogni speranza, voi  
ch'intrate."*

---

— Dante Alighieri,  
*Divina Commedia*

With the evolution of technologies brought in by the new millennium and the exponential growth of Internet-based services targeting millions of users all over the world, the Software Engineers' community has been continuously tested by an ever growing number of challenges related to management of increasingly large amounts of user data<sup>[8]</sup>.

This phenomena is commonly referred to as Big Data. A very popular 2001 research report<sup>[14]</sup> by analyst Doug Laney, proposes a definition of big data based on its three defining characteristics:

- *Volume*: the quantity of data applications have to deal with, ranging from small - e.g. locally maintained Databases - to large - e.g. distributed File Systems replicated among data centers.
- *Variety*: the type and structure of data, ranging from classic SQL-structured data sets to more diversified and unstructured ones such as text, images, audio and video.
- *Velocity*: the speed at which data is generated, establishing the difference between pull-based systems, where data is synchronously pulled by the consumer, and push-based systems, more suited for handling real-time data by asynchronously pushing it to its clients.

Each of these traits directly influences the way programming languages, APIs and databases are designed today. The increasing volume calls for a declarative approach to data handling as opposed to an imperative one, resulting in the developer's focus shifting from how to compute something to what it is to be computed in the first place<sup>[7]</sup>. The diversification of data, on the other hand, is the main drive for the research and development of noSQL approaches to data storage. Lastly, the increase in velocity fuels the need for event-driven, push-based models of computation that can better manage the high throughput of incoming data<sup>[16]</sup>.

In this context, the concept of *reactive programming* has gained much traction in the developer's community as a paradigm well-suited for the development of asynchronous event-driven applications<sup>[3]</sup>. Unfortunately, reactive programming has been at the center of much discussion, if not confusion, with regards to its definition, properties and principles that identify it<sup>[17]</sup>.

The goal of our work is to use mathematics as a tool to analyze today's commercial reactive libraries and understand how they relate to the theoretical concept of reactive programming. We are going to do so by utilizing concepts and ideas from functional programming and category theory with the purpose of deriving the reactive types, and will subsequently continue with the development of a reference reactive library which builds upon the previously derived theoretical foundations.

## Contributions

To the best of our knowledge, we are not aware of any previous work which analyses reactive programming from a theoretical standpoint or derives its types and interfaces through the use of mathematics.

The most commonly known attempt at defining reactive programming is the Reactive Manifesto<sup>[13]</sup>, a document that tries to describe reactive systems in terms of design principles and conceptual architecture. Although certainly insightful, we feel the document targets a more managing-focused audience as opposed to software developers and engineers.

Our work, on the other hand, aims at providing types and interfaces that describe the real essence of the reactive paradigm, aiding engineers that wish to use or develop reactive libraries in understanding and taking more informed decisions on the matter.

This goal is achieved by providing a mathematical derivation of the reactive types, starting from their interactive counterparts and making use of theoretical concepts from category theory. These derived types are then used in the implementation of a formal reactive library where the purpose is showing how the theoretical definitions given to the various components can effectively be translated into working code.

Together with this formal definition of the paradigm, this work presents a reference implementation for a production level reactive library, as well as a highlight of the issues and challenges encountered when bridging from the theoretical foundations of reactive programming to a concrete implementation of a reactive API.

With the help of this report and the associated code repository, any software engineer interested in the topic should be able to understand the theoretical foundations behind the reactive paradigm and develop a reactive library in any language of choice.

---

## Overview

Chapter 1 introduces the scope of our research, providing a definition of reactive programming, the motivation and reasoning behind our research and an overview of the current technologies and APIs that claim to belong to the world of reactive programming. Chapter 2 presents the mathematical derivation of the reactive types and interfaces, starting from the definition of `Iterable` and ending with that of `Observable`. Chapter 3 builds the bridge between the formal definition of the reactive types and a production level implementation of the paradigm, highlighting the technical issues as well as analyzing the relations with the previously discussed formal definitions. Chapter 4 concludes with final thought and future work.

## Notation & Conventions

In the exposition of our work we will make use of Haskell as the reference language. This decision is motivated by the language's strong connection with mathematics as well as it's clean syntax.

All the code presented in this report, a minimal complete theoretical implementation and a reference implementation of a reactive library can be found at the associated code repository on Github - <https://github.com/Widar91/Thesis>.

Fix repo organization



## REACTIVE PROGRAMMING

*The cold winds are rising in the North... Brace yourselves, winter is coming.*

---

— George R.R. Martin,  
*A Game of Thrones*

In this chapter we are going to introduce the concept of reactive programming and motivate its importance and relevance with regards to modern applications and the type of problems developers have to face nowadays. We are then going to introduce the most popular commercial libraries that claim to solve the reactive problem, with the purpose of giving the reader some context for our discussion and motivating the need for a mathematical formalization that abstracts over the class of problems these implementations set out to address.

## 1.1 The Essence of Reactive Programs

The use of the term reactive program in scientific literature is dated back to the mid-sixties<sup>[1]</sup>. A relevant and insightful definition was given by G. Berry in 1991<sup>[4]</sup> as he describes reactive programs in relation to their dual counterparts, interactive programs:

*“Interactive programs interact at their own speed with users or with other programs; from a user point of view, a time-sharing system is interactive. Reactive programs also*

maintain a continuous interaction with their environment, but at a speed which is determined by the environment, not by the program itself.”

Interactive programs concretize the idea of a pull-based model of computation, where the program - the consumer in this case - has control over the speed at which data will be requested and handled. A perfect example of an interactive program is a control-flow structure such as a for-loop iterating over a collection of data: the program is in control of the speed at which data is retrieved from the containing collection and will request the next element only after it is done handling the current one.

Reactive programs, on the contrary, embody the idea of a push-based - or event-driven - model of computation, where the speed at which the program interacts with the environment is determined by the environment rather than the program itself. In other words, it is now the producer of the data - i.e. the environment - who determines the speed at which events will occur whilst the program's role reduces to that of a silent observer that will react upon receiving events. Standard example of such systems are GUI applications dealing with various events originating from user input - e.g. mouse clicks, keyboard button presses - and programs dealing with stock markets, social media or any other kind of asynchronous updates.

## 1.2 Why Reactive Programming Matters

Considering the definition and examples of reactive programs we analyzed in the previous section, let's now try to formalize the class of problems the reactive programming paradigm is specifically well-suited for.

The table below provides a collection of types offered by common programming languages for handling data, parameterized over two variables: the size of the data, either one or multiple values, and the way data is handled, either by synchronous or asynchronous computations<sup>[18]</sup>.

	One	Many
Sync	<code>a</code>	<code>Iterable a</code>
Async	<code>Future a</code>	<i>Reactive Programming</i>

The first row shows that synchronous functions come in two flavors: classic functions that return a single value of type `a` and functions that produce a collection of results of type `a`, abstracted through the `Iterable a` interface (See section 2.1). These types of functions embody the standard imperative, pull-based approach to programming, where a call to a function/method synchronously blocks until a result is produced.

Moving on to the second row, we encounter `Future a`, an interface representing an asynchronous computation that, at a certain point in the future, will result in a value of type `a`. Futures are generally created by supplying two callbacks together with the asynchronous computation, one to be executed in case of success and the other one in case of error.

Programming languages, however, are not as well equipped when it comes to handling asynchronous computations resulting in multiple values - i.e. push-based collections. The issue lies in the fact that the program's control flow is dictated by the environment rather than the program itself - i.e. inversion of control -, making it very hard to model such problems with commonly known control structures, which are optimized for sequential models of computation. Traditional solutions typically involve developers manually trying to compose callbacks by explicitly writing CPS (continuation passing style) code<sup>[18]</sup>, resulting in what it's commonly referred to as *Callback Hell*<sup>[6]</sup>.

The aforementioned class of problems reflects the definition of reactive programs we analyzed in the previous section, where the environment asynchronously - i.e. at its own speed - pushes multiple events to the program. The reactive programming paradigm sets out to provide interfaces and abstractions to facilitate the modeling of such problems as push-based collections.

## 1.3 Related Work

Interfaces are only as good as the implementations that back them up. In this section we are going to discuss and analyze various APIs and libraries that claim to embody the reactive paradigm, motivating our need for a mathematical formalization to aid in unifying these different approaches under a single set of interfaces, eliminating doubts and confusion when it comes to the essence of reactive programming.

Talk about FRP, Reactive Streams, Rx and other stuff. Cite survey paper Claim that Observable is dual of Iterator in order to justify the why.

### 1.3.1 Reactive Extensions

### 1.3.2 Reactive Streams

Try to define itself as reactive programming, mathematics prove the claim wrong, they implement asynchronous iterables. This does not make them useless, yet they are not a good solution for the problem we are trying to solve.

Probably put the derivation of Reactive Streams in the appendix, it's not really the focus, I guess.

### **1.3.3 Functional Reactive Programming**

Much researched topic, beautiful in theory, yet based on continuous time, un-achievable with computers where time is inherently discrete. Connel Elliot, the inventor of FRP, says that commercial implementations are far from the original theory and do not really represent the essence of FRP



## INTO THE RABBIT HOLE: *Deriving the Observable*

*"It was much pleasanter at home," thought poor Alice, "when one wasn't always growing larger and smaller, and being ordered about by mice and rabbits. I almost wish I hadn't gone down the rabbit-hole – and yet – and yet – ..."*

---

— Lewis Carol,  
*Alice in Wonderland*

As we saw in Chapter 1, the `Iterable` interface embodies the idea of a pull-based model of computation and is the commonly adopted solution to dealing with synchronous computations resulting in multiple values. In this chapter we are going to formalize the intuition that there exists a duality relation between interactive and reactive programs, as well as between pull and push models of computations, by deriving the `Observable` interface starting from its dual counterpart, the `Iterable`.

The derivation that follows will require the use of a number mathematical concepts such as *categorical duality*, *continuations*, *(co)products*, *(un)currying*, *covariance*, *contravariance* and *functors*. We suggest the reader to get familiar with these topics before diving into the derivation. An accessible introduction to each can be found in Appendix A.

## 2.1 Iterables

An `Iterable` is a programming interface which enables the user to traverse a collection of data, abstracting over the underlying implementation<sup>[9]</sup>.

The interface and semantics of `Iterable`s were first introduced by the Gang of Four through their Iterator pattern<sup>[9]</sup>; today's most used programming languages introduce the `Iterable` as the root interface for standard pull collections APIs - exposing concrete implementations such as maps, sets, indexed sequences and so on.

The `Iterable` interface is generally fixed across programming languages, with the exception of naming conventions - C# and related languages call it `IEnumerable` - and slight differences in the types, as we can see from the following definitions.

---

```
1  -- Java Iterable
2  newtype Iterable a = Iterable
3      { getIterator :: () -> IO (Iterator a)
4      }
5
6  data Iterator a = Iterator
7      { hasNext :: () -> Bool
8      , next    :: () -> IO a
9      }
10
11  -----
12
13  -- C# IEnumerable
14  newtype IEnumerable a = IEnumerable
15      { GetEnumerator :: () -> IO (IEnumerator a)
16      }
17
18  data IEnumerator a = IEnumerator
19      { moveNext :: () -> IO Bool
20      , current  :: () -> a
21      }
```

---

Although the essence of the pattern is preserved by both definitions, we claim that the C# version

more clearly and accurately reflects the way side effects play a role in the usage of the interface: `moveNext` contains all the side effects of traversing the underlying collection and retrieving the next value while `current` can inspect the retrieved value multiple times in a pure way. The Java version, on the other hand, embeds the side effect in the `next` function, making it impossible to inspect the current value multiple times. For this reason and without loss of generality, we will make use of the C# definition - modulo naming conventions - in the reminder of the discussion:

---

```
1 newtype Iterable a = Iterable
2   { getIterator :: () -> IO (Iterator a)
3   }
4
5 data Iterator a = Iterator
6   { moveNext :: () -> IO Bool
7   , current  :: () -> a
8   }
```

---

## 2.2 The Essence of Iterables

The first step in deriving the `Observable` is to simplify our `Iterable` definition to a type that reflects its very essence; we are gonna do this by stripping the interface presented in the previous section of all the unnecessary operational features that only clutter our definition.

Let's start by taking a closer look at the `Iterator` interface; we can observe that the definition of the functions `moveNext` and `current` is equivalent to a single function which returns either a value - analogous to a `moveNext` call returning true and a subsequent invocation to `current` - or nothing - analogous to a call to `moveNext` returning false.

Before we formalize this observation with a proper type, let us notice another effect that is hidden in the current definition of `moveNext` and not made explicit by its type: the possibility for an exception to be thrown by the function's body.

By merging these considerations with the notion of coproducts and Haskell's `Either` and `Maybe` type, we obtain the following definition.

---

```
1 newtype Iterable a = Iterable
2   { getIterator :: () -> IO (Iterator a)
```

```
3     }  
4 newtype Iterator a = Iterator  
5     { moveNext :: () -> IO (Either SomeException (Maybe a))  
6     }
```

---

Note how, theoretically, `getIterator` could also throw an exception. We assume here, without loss of generality, that the function will never throw and will always be able to return an `Iterator` instance.

The next step is to forget about data types and express our interfaces as simple types.

---

```
1 type Iterable a = () -> IO (Iterator a)  
2 type Iterator a = () -> IO (Either SomeException (Maybe a))
```

---

At this point, we want to put aside the operational concerns regarding exceptions and termination and assume the `Iterator` function will always return a value of type `a`.

---

```
4 type Iterable a = () -> IO (() -> IO a)  
5 type Iterator a = () -> IO a
```

---

We have now reached a point where no simplification is possible anymore. The obtained types reflect the essence of the Iterator pattern: an `Iterable` is, theoretically, a function which, when invoked, produces an `Iterator` and an `Iterator` is itself a function producing a value of type `a` as a side effect.

When looking at the `Iterator` type from an object oriented perspective, the reader should notice a strict similarity to a *getter* function - i.e. a lazy producer of values: iterators are, in fact, nothing more than getters of values of type `a`. The `Iterable`, on the other hand, is a function that enables the user to get an `Iterator`, i.e. a getter of a getter of `a`.

This correspondence will turn out to be very insightful later on in our discussion, where we will observe that `Observable` is nothing more than a setter of setters, another instance of duality in our formalization.

When looking at the relation between the `Iterator` type and its base component, `a`, we can observe how they are bound by a covariant relation:

---

```
1           A <: B
2           -----
3           () -> IO A <: () -> IO B
```

---

The intuition can be easily understood when we think of an iterator as a drink vending machine, i.e. a function which, whenever called, will give back a drink:

---

```
1           Coke <: Drink
2           -----
3           VendingM Coke <: VendingM Drink
```

---

If coke is a subtype of drink, then whenever I am asked for a drink vending machine, I can hand out a coke vending machine without incurring in any troubles with the person who asked, as that machine will correctly provide drinks - even though they will always be coke - whenever prompted for one.

With `Iterator` being a getter itself, it should be clear how covariance plays the same role as with `Iterable`.

To formally prove the intuition of a covariant relation, we instantiate the `Iterable` / `Iterator` types to a covariant `Functor` and prove its laws in the following snippet.

---

```
1 import Control.Monad
2
3 {-
4    Using a type synonym instead of Haskell's newtypes,
5    in order to avoid clutter in our proofs:
6
7    type Iterator a = () -> IO a
8
```

```
9     fmap :: (a -> b) -> Iterator a -> Iterator b
10     fmap f ia = \() -> ia () >>= return . f
11
12     identity:
13         fmap id
14             -- eta abstraction
15     = \ia -> fmap id ia
16         -- definition of fmap
17     = \ia -> \() -> ia () >>= return . id
18         -- application of id
19     = \ia -> \() -> ia () >>= return
20         -- IO monad right identity*
21     = \ia -> \() -> ia ()
22         -- eta reduction
23     = \ia -> ia
24         -- definition of
25     = id
26
27     composition:
28         (fmap p) . (fmap q)
29             -- eta abstraction
30     = \ia -> ((fmap p) . (fmap q)) ia
31         -- definition of (.)*
32     = \ia -> fmap p (fmap q ia)
33         -- definition of fmap q
34     = \ia -> fmap p (\() -> ia () >>= return . q)
35         -- definition of fmap p
36     = \ia -> \() -> (\() -> ia () >>= return . q) () >>= return . p
37         -- eta reduction inner lambda
38     = \ia -> \() -> ia () >>= return . q >>= return . p
39         -- eta abstraction
40     = \ia -> \() -> ia () >>= \a -> (return . q) a >>= return . p
41         -- definition of (.)
42     = \ia -> \() -> ia () >>= \a -> return (q a) >>= return . p
43         -- IO monad left identity*
44     = \ia -> \() -> ia () >>= \a -> (return . p) (q a)
45         -- definition of (.)
46     = \ia -> \() -> ia () >>= \a -> (return . p . q) a
```

```
47         -- definition of (.)
48     = \ia -> \() -> ia () >>= \a -> return ((p . q) a)
49         -- definition of fmap
50     = \ia -> fmap (p . q) ia
51         -- eta reduction
52     = fmap (p . q)
53
54 * monad right identity:
55     m >>= return = m
56
57 monad left identity:
58     return a >>= f = f a
59
60 definition of (.):
61     (.) :: (b -> c) -> (a -> b) -> a -> c
62     (f . g) a = f (g a)
63 -}
64
65 newtype Iterator a = Iterator { moveNext :: () -> IO a }
66
67 instance Functor Iterator where
68     fmap f ia = Iterator $ \() -> liftM f (moveNext ia ())
69
70 {-
71     type Iterable a = () -> IO (Iterator a)
72
73     fmap :: (a -> b) -> Iterable a -> Iterable b
74     fmap f iia = \() -> iia () >>= return . fmap f
75
76     identity:
77         fmap id
78         -- eta abstraction
79     = \iia -> fmap id iia
80         -- definition of fmap
81     = \iia -> \() -> iia () >>= return . fmap id
82         -- Iterator identity law
83     = \iia -> \() -> iia () >>= return . id
84         -- application of id
```

```
85     = \ia -> \() -> ia () >=> return
86         -- IO monad right identity
87     = \ia -> \() -> ia ()
88         -- eta reduction
89     = \ia -> ia
90         -- definition of id
91     = id
92
93 composition:
94     (fmap p) . (fmap q)
95         -- eta abstraction
96     = \ia -> ((fmap p) . (fmap q)) ia
97         -- definition of (.)
98     = \ia -> fmap p (fmap q ia)
99         -- definition of fmap
100    = \ia -> fmap p (\() -> ia () >=> return . fmap q)
101        -- definition of fmap
102    = \ia -> \() -> (\() -> ia () >=> return . fmap q) () >=> return . fmap p
103        -- eta reduction
104    = \ia -> \() -> ia () >=> return . fmap q >=> return . fmap p
105        -- eta abstraction
106    = \ia -> \() -> ia () >=> \ia -> (return . fmap q) ia >=> return . fmap p
107        -- definition of (.)
108    = \ia -> \() -> ia () >=> \ia -> return (fmap q ia) >=> return . fmap p
109        -- IO monad left identity
110    = \ia -> \() -> ia () >=> \ia -> (return . fmap p) (fmap q ia)
111        -- definition of (.)
112    = \ia -> \() -> ia () >=> \ia -> (return . fmap p . fmap q) ia
113        -- eta reduction
114    = \ia -> \() -> ia () >=> return . fmap p . fmap q
115        -- Iterator composition law
116    = \ia -> \() -> ia () >=> return . fmap (p . q)
117        -- definition of fmap
118    = \ia -> fmap (p . q) ia
119        -- eta reduction
120    = fmap (p . q)
121 -}
```



```
123 newtype Iterable a = Iterable { getIterator :: () -> IO (Iterator a) }
124
125 instance Functor Iterable where
126     fmap f iia = Iterable $ \() -> liftM (fmap f) (getIterator iia ())
```

---

For the sake of completeness, it is worth mentioning that `Iterable` is, among others, also an instance of Applicative Functor and Monad. Although certainly interesting from a theoretical perspective, showing these instances and proving the associated laws goes beyond the scope of this work. Nonetheless, we will see in the next section how these concepts are relevant in expressing and motivating the duality between Iterables and Observables.

## 2.3 Applying Duality

By now, the reader should be somehow familiar with the concept of duality, as it has appeared many times throughout our discussion in concepts such as pull and push models of computation or interactive and reactive programs. Duality is, in fact, a very important general theme that has manifestations in almost every area of mathematics<sup>[10]</sup> (See Appendice A for an introductory discussion on the topic).

Starting from the fact that the `Iterable` interface embodies the idea of interactive programming, let's use the principle of duality to derive the `Observable` interface and see how it relates to the concept of reactive programming. In practice, this translates to the simple task of flipping the function arrows in the `Iterable` interface, taking us from a function resulting in a value of type `a` to one accepting an `a`.

---

```
1 {-
2     () -> (() -> a) -- iterable
3     () <- (() <- a) -- apply duality
4     (a -> ()) -> () -- observable
5 -}
6
7 type Iterator a = () -> IO a
8     -- = () IO <- a
9 type Observer a = a -> IO ()
10
11 type Iterable a = () -> IO (() -> IO a)
```

```
12         -- = () IO <- (() IO <- a)
13 type Observable a = (a -> IO ()) -> IO ()
```

---

Note how the side effects are bound to function application rather than values, hence their flipped position in the `Observable` type.

The newly derived types are relatively easy to read and understand: `Observer` is simply a function that, given a value of type `a` will handle it somehow, producing side effects; the `Observable`, on the other hand, is responsible for producing such values of type `a` and feeding them to the `Observer` it has been given as an argument.

In the previous section we have discussed many properties associated with `Iterable` s. Let's analyze now how these properties translate under dualisation and how they affect our new derived interface, the `Observable`.

First, moving from the observation that an `Iterable` is a getter of a getter, we can observe that the `Observable` plays exactly the opposite role, that is, a setter of a setter. The type `Observer :: a -> IO ()` represents, in fact, the essence of a setter function, whereas the `Observable` consists in nothing more than the simple task of applying the observer function to itself, producing a setter of setters.

While the discussion about `Iterable`'s covariance was quite intuitive, things get a little bit more complicated when analyzing `Observable` s. Referring back to our previous example involving cokes and drinks, we can now think of the `Observer` as a recycling machine:

---

```
1             Coke <: Drink
2             -----
3             RecyclingM Drink <: RecyclingM Coke
```

---

Our intuition tells us that this time, a recycling machine that can only handle coke cannot be used in place of one that needs to handle any type of drinks, as it would fail at its task whenever a drink that is not a coke is fed into it. On the other hand, a recycling machine that works for any type of drink can be safely used in place of one that needs to handle cokes. This intuition bounds `Observer` and its base type `a` by a contravariant relation:

---

```

1           A <: B
2           -----
3           A -> IO () <: B -> IO ()

```

---

A more theoretical take on the matter involves the notion of type's positivity and negativity: we can interpret a function of type  $f :: a \rightarrow b$  as a way for us to produce a value of type  $b$ . In this context,  $b$  is considered to be positive with respect to the type  $a \rightarrow b$ . On the other hand, in order to apply the function, we are going to need a value of type  $a$ , which we will need to get from somewhere else;  $a$  is therefore considered to be negative w.r.t. the function type, as the function introduces a need for this value in order to produce a result. The point of this distinction is that positive type variables introduce a covariant relation between base and function type whereas negative type variables introduce a contravariant relation.

Analyzing `Iterable` within this framework is easy, the `Iterator` function contains a single type parameter found in a positive position, therefore resulting in a covariant relation; being the `Iterable` the result of applying the `Iterator` function to itself, we again result in a covariant relation w.r.t. the type parameter `a`.

The `Observer` function, on the contrary, introduces a need for a value of type `a`, resulting in a contravariant relation w.r.t. `a`. Again, the `Observable` function is the result of applying `Observer` to itself; surprisingly, this results in `a` being in a positive position. The intuition is easily understood by thinking about the rules of arithmetic multiplication: `a` is in negative position w.r.t. the `Observer` function, whereas the `Observer` is in negative position w.r.t. the `Observable`. This leads to `a` being negated twice, ultimately resulting in a positive position within the `Observable` function.

---

```

3  f    ::  a ->  b
4      = -a -> +b
5
6  g    ::  ( a ->  b) ->  c
7      = -(-a -> +b) -> +c
8      =  (+a -> -b) -> +c
9
10 observer
11     ::  a ->  ()
12     = -a ->  ()
13
14 observable

```

```
15     :: ( a -> () ) -> ()
16     = -(-a -> () ) -> ()
17     = (+a -> () ) -> ()
```

---

Before we formalize this claim, let's convince ourselves that `Observable` s effectively produces a value of type `a` by looking at an example:

```
21 randomValueObs :: Observable Int
22 randomValueObs = Observable $ \observer -> do
23     int <- randomRIO (1, 10)
24     observer int
```

---

It is clear from this implementation that `randomValueObs` indeed produces a value of type `Int` , whereas the `Observer` introduces a need for such value in order to be applied. For more details on the positivity and negativity of functions and type variables, see<sup>[22] [11]</sup>.

Just as we did with `Iterable/Iterator` , we can formally prove the covariant and contravariant relations between `Observable/Observer` and their base type `a` by instantiating them to `Functor` and `Contravariant (Functor)` respectively, as well as proving the laws.

```
1 import Data.Functor.Contravariant
2
3 {-
4     type Observer a = a -> IO ()
5
6     contramap :: (a -> b) -> Observer b -> Observer a
7     contramap f ob = ob . f
8
9     identity:
10         contramap id
11         = \ob -> contramap id ob
12         = \ob -> ob . id
13         = \ob -> ob
14         = id
```

```

15
16 composition:
17     (contramap p) . (contramap q)
18     = \ob -> ((contramap p) . (contramap q)) ob
19     = \ob -> contramap p (contramap q ob)
20     = \ob -> contramap p (ob . q)
21     = \ob -> (ob . q) . p
22     = \ob -> ob . (q . p)
23     = \ob -> contramap (q . p) ob
24     = contramap (q . p)
25 -}
26
27 newtype Observer a = Observer { onNext :: a -> IO () }
28
29 instance Contravariant Observer where
30     contramap f ob = Observer $ onNext ob . f
31
32 {-
33     type Observable a = Observer a -> IO ()
34
35     fmap :: (a -> b) -> Observable a -> Observable b
36     fmap f ooa = \ob -> ooa (contramap f ob)
37
38     identity:
39         fmap id
40         = \ooa -> fmap id ooa
41         = \ooa -> \ob -> ooa (contramap id ob)
42         = \ooa -> \ob -> ooa ob
43         = \ooa -> ooa
44         = id
45
46     composition:
47         fmap p . fmap q
48         = \ooa -> (fmap p . fmap q) ooa
49         = \ooa -> fmap p (fmap q ooa)
50         = \ooa -> fmap p (\ob -> ooa (contramap q ob))
51         = \ooa -> \oc -> (\ob -> ooa (contramap q ob)) (contramap p oc)
52         = \ooa -> \oc -> ooa (contramap q (contramap p oc))

```

```

53     = \ooa -> \oc -> ooa ((contramap q . contramap p) oc)
54     = \ooa -> \oc -> ooa (contramap (p . q) oc)
55     = \ooa -> fmap (p . q) ooa
56     = fmap (p . q)
57 -}
58
59 newtype Observable a = Observable { subscribe :: Observer a -> IO () }
60
61 instance Functor Observable where
62     fmap f ooa = Observable $ subscribe ooa . contramap f

```

---

The reader acquainted with functional programming will easily see the resemblance between the `Observable` type and a CPS function (See Appendice A).

---

```

1 cont      :: (a -> r    ) -> r
2 observable :: (a -> IO ()) -> IO ()

```

---

It is clear how an `Observable` is nothing more than a special case of a CPS function where the result type `r` is instantiated to `IO ()`. To convince ourselves of this equivalence, let's think about the definition of a CPS function, i.e. a suspended computation which, given another function - the continuation - as argument, will produce its final result. This definition suits perfectly the idea behind `Observable` discussed in Section 1.3.1: a function which will do nothing - i.e. is suspended - until it is subscribed to by an `Observer`.

A continuation, on the other hand, represents the future of the computation, a function from an intermediate result to the final result<sup>[19]</sup>; in the context of `Observable`s, the continuation represents the `Observer`, a function specifying what will happen to a value produced by the `Observable`, whenever it will become available, that is, whenever it will be pushed into the `Observer`. Since a continuation can be called multiple times within the surrounding CPS context, it is easy to see how this mathematical concept allows us to deal with multiple values produced at different times in the future.

We can prove our claim by implementing the `Observable` interface using Haskell's Continuation Monad Transformer and observing how the unwrapping function `runContT` effectively hands us back our original type:

---

```
1 newtype ContT r m a :: * -> (* -> *) -> * -> *
2 runContT :: ContT r m a -> (a -> m r) -> m r
3
4 type Observable a = ContT () IO a
5 runContT :: ContT () IO a -> (a -> IO ()) -> IO ()
```

---

This equivalence is very important as it allows us to claim an instance of `Applicative Functor` and `Monad` for our derived type, `Observable`. These instances are inherited for free from the continuation monad, sparing us the burden of implementing them and proving all related laws.

---

```
1 instance Applicative (ContT r m) where
2     pure x = ContT ($ x)
3     f <*> v = ContT $ \c -> runContT f $ \g -> runContT v (c . g)
4
5 instance Monad (ContT r m) where
6     return x = ContT ($ x)
7     m >>= k = ContT $ \c -> runContT m (\x -> runContT (k x) c)
```

---

## 2.4 Termination and Error Handling

We began this chapter by progressively simplifying the `Iterable`'s interface in order to derive a type that would theoretically represent its very essence. One of the most important steps was setting aside concerns regarding termination and error handling of a collection. We are now going to reshape our reactive interfaces in order to address these concerns and appropriately describe the potential side effects directly in the types.

Informally, an `Observable` stream might not only produce one or more values, but it might gracefully terminate at a certain point in time or throw an exception and abruptly terminate whilst processing values. A more appropriate type for `Observer` is then the following:

---

```
1 newtype Observer a = Observer
2     { onNext :: Either SomeException (Maybe a) -> IO ()
3     }
```

Just as with `Iterable`, the introduction of `Either SomeException` allows us to express that the `Observer` can handle unexpected exceptions, while the `Maybe` reflects the possibility for a stream to end and propagate no more values.

Unfortunately, this type is very hard to read as well as understand for someone new to the topic. Looking at the matter from a functionality point of view, what we really need is for our CPS function - i.e. the `Observable` - to accept two additional continuations, one dealing with completion and one with exceptions. We can achieve this by first noticing that our type is nothing more than a coproduct - the same that we introduced previously for `Iterable` - of three base types: `a + SomeException + ()`. By utilizing the notion of product - the dual of coproduct - we can split the function handling the initial type into three different ones. This brings us to the final version of our reactive interfaces for push-base collections<sup>1</sup>

---

```
1 newtype Observable a = Observable
2   { subscribe :: Observer a -> IO ()
3   }
4
5 data Observer a = Observer
6   { onNext      :: a -> IO ()
7   , onError     :: SomeException -> IO ()
8   , onCompleted :: IO ()
9   }
```

---

The `Observable` is now a special version of a CPS function accepting three continuation functions - embedded inside the `Observer` -, one for each effect an `Observable` can propagate: value, termination or exception.

## 2.5 Formalizing Observables

In section 2.3 we have shown how the essential type for `Observable` is effectively nothing more than a particular instance of the continuation monad. In this section we are going to explore this relation in further detail, introducing a notation which will help us keep track of the changes we will

---

<sup>1</sup>The two definitions are equivalent also from an implementation point of view, the first simulating the second though the use of pattern matching.



make to the original `Observable` type, ultimately showing how the resulting interface - that will be used in our final library - will consist in nothing more than a modified version of a CPS function.

We are going to start from the notion that `Observable` is, at its essence, nothing more than a setter of setters, the result of applying the `Observer` function to itself. We can then express the `Observer` as a function `(!)`<sup>2</sup> that negates its type argument and results in a side-effectful computation.

---

```
1 !a :: a -> IO ()
```

---

When we apply the function to itself - i.e. substitute `a` for `!a` - we obtain our first definition of `Observable`, a CPS function that instantiates the result to `IO ()`.

---

```
1 !!a :: (a -> IO ()) -> IO ()
```

---

As we have seen in the previous section, this definition is not expressive enough when we want to make explicit all the effects that are involved when dealing with push-base collections. It is therefore necessary to deviate from the standard definition of continuation and replace the inner application of `(!)` with a new function `(?)`, whose type embed the involved effects:

---

```
1 ?a :: Either Error (Maybe a) -> IO () -- termination and error handling
2 !?a :: (Either Error (Maybe a) -> IO ()) -> IO ()
```

---

Note how this definition is equivalent to the one used in the previous section, where we used the notion of products to unwrap the `(?)` function into three different continuation, each addressing one of the possible effects.

In the next chapter we are going to further modify this definition with the inclusion of a cancellation mechanism.

This newer version of `Observable` is still implementable as an instance of the continuation monad, as the code below shows.

---

<sup>2</sup>Regard the code used in this explanation as pseudo-Haskell.

```
5  -- Event a = Either SomeException (Maybe a)
6  data Event a = OnNext a | OnError SomeException | OnCompleted
7      deriving Show
8
9  type Observer a = Event a -> IO ()
10 type Observable a = ContT () IO (Event a)
11
12 newObservable :: (Observer a -> IO ()) -> Observable a
13 newObservable = ContT
14
15 subscribe :: Observable a -> Observer a -> IO ()
16 subscribe = runContT
```

---

The code above uses a slightly different approach to expressing the three types of side effects an `Observer` has to deal with. Instead of using `Either` and `Maybe` from Haskell's libraries, we utilize our own custom datatype `Event`, a coproduct of values of type `a + SomeException + ()`; although the two definitions are equivalent in every aspect, the adopted one offers more clarity in terms of code readability.

At this point we have all the necessary tools to create and run an `Observable`.

---

```
18 obs = newObservable $ \observer ->
19     do observer (OnNext 1)
20        observer (OnNext 2)
21        observer OnCompleted
22
23 main :: IO ()
24 main = subscribe obs print
25
26 {-
27 output>
28     OnNext 1
29     OnNext 1
30     OnCompleted
31 -}
```

Notice how, being a CPS function, an `Observable` only pushes values once subscribed to and acts as a suspended computation otherwise.

The code above, being a toy example, fails to show some fundamental properties associated with this new interface; in particular, it fails to show how `Observable`s can actually handle asynchronous sources of data. The following snippet of code contains a more realistic and meaningful example of an `Observable` handling keyboard presses, asynchronous events by nature: whenever the user presses a key, an event containing the corresponding character is propagated to the `Observer` and will eventually be printed on the command line. It is worth noticing how our basic implementation of `Observable` based on continuations works just as well as a full blown one in terms of its core capability of handling asynchronous data.

---

```
17 obs :: Observable Char
18 obs = newObservable $ \observer -> do
19     keyboardCallback $= Just (\c p -> observer (OnNext c))
20
21 display :: DisplayCallback
22 display = do
23     clear [ ColorBuffer ]
24     flush
25
26 main :: IO ()
27 main = do
28     (_progName, _args) <- getArgsAndInitialize
29     _window <- createWindow "Observable Keyboard"
30     subscribe obs (print . show)
31     displayCallback $= display
32     mainLoop
```

---

This example brings us to the following observation: `Observable`s are capable of handling asynchronous data sources, yet the means by which the data is handled are not asynchronous by default. This is a common misconception and source of much confusion among the community: the `Observable` interface is not opinionated w.r.t. concurrency and therefore, by default, synchronously handles its incoming data, blocking the next incoming events whilst processing the current one. This

behavior is not fixed though: as we will see in Chapter 3, it is possible to make use of `Scheduler`s to orthogonally introduce concurrency in our reactive systems, altering the control flow of the data processing allowing the user to dispatch the work to other threads.

At this point in the discussion we have arrived to a working implementation of a push based collection purely derived from mathematical and categorical concepts such as duality and continuations. In spite of being very insightful for theoretical discussions on the properties and relations of `Observable` and the continuation monad, this implementation of the reactive types is impractical in the context of a full fledged API. In the next Chapter we will take the necessary steps to build the bridge between theory and practice, providing a reference implementation of `Observable` more adapt to be utilized in real world applications.

For a full implementation of a Reactive Library based on the ideas presented in this section, aimed at highlighting the strong connection between `Observable` and other already existing functional structures from which it composes, see Appendice B.

## OUT OF THE RABBIT HOLE: *Towards a usable API*

*"In theory there is no difference  
between theory and practice; in  
practice there is."*

---

— Nassim Nicholas Taleb,  
*Antifragile - Things that Gain From  
Disorder*

So far we focused our analysis on the essence of the `Observable` interface, setting aside the many operational concerns that would come up when trying to implement these concepts into a usable, commercial API. In this chapter we are going to build the bridge between our theoretical definition of `Observable` and a concrete and usable implementation of a reactive library, to which we will refer to as Rx.

The goal of this chapter is to provide the reader with a reference implementation of a reactive library as well as to highlight the challenges and issues that emerge when trying to build the bridge between theory and practice. The implementation choices presented below are in no way prescriptive, instead, they aim to describe the problem in the most clear way, in order to stimulate awareness rather than blindly guide the reader to a solution.

In the remainder of the discussion, we are going to introduce the *Reactive Contract*, a set of assumptions on the reactive types our library is going to build upon, *Schedulers*, which will allow us to bring concurrency into our reactive equation, *Subscriptions*, used to implement a mechanism for

Say something about why I chose to go with the RxJava style of implementation?

premature stream cancellation and finally, *Operators*, the means with which we will make our reactive streams composable.

For the sake of clarity and completeness, the following set of interfaces represents the starting point for our discussion:<sup>1</sup>

---

```
1 newtype Observable a = Observable
2   { _subscribe :: Observer a -> IO ()
3   }
4 data Observer a = Observer
5   { onNext      :: a -> IO ()
6   , onError     :: SomeException -> IO ()
7   , onCompleted :: IO ()
8   }
```

---

It is worth noting that even though the `Observable`’s theoretical foundations lie in the realm of functional programming, the road to making it usable is full of obstacles that are often better tackled using imperative programming features, such as state. As much as I personally prefer a functional and pure approach to programming, I will favor, in the rest of the discussion, the solution that most clearly and easily solves the problem, be that functional or imperative.

## 3.1 The Reactive Contract

The `Observable` and `Observer` interfaces are somewhat limited, in their expressive power, to only argument and return types of their functions. The reactive library we are going to build is going to make more assumptions than the ones expressible by the type system. Although limiting, in a sense, the freedom with which the reactive interfaces can be utilized, this set of assumptions - the *Contract* - greatly facilitates reasoning about and proving correctness of reactive programs<sup>[2]</sup>.

In later sections, we will refer back to these assumptions when discussing the actual implementation of our reactive library.

### 3.1.1 Reactive Grammar

The first assumption we are going to introduce involves restrictions on the emission protocol of an `Observable`. Events propagated to the `Observer` continuation will obey the following regular

---

<sup>1</sup> `subscribe` has been renamed to `_subscribe` in order to avoid naming conflicts later on in the discussion and reflect the fact that it should not be used directly by the user.

expression:

```
onNext* (onError | onCompleted)?
```

This grammar allows streams to propagate any number - 0 or more - of events through the `onNext` function, optionally followed by a message indicating termination, be that natural - through `onCompleted` - or due to a failure - through `onError`. Note how the optional nature of a termination message allows for the existence of infinite streams.

This assumption is of paramount importance as it guarantees that no events can follow a termination message, allowing the consumer to effectively determine when it is safe to perform resource cleanup operations.

### 3.1.2 Serialized Messages

Later in this chapter we will see how we can introduce concurrency in our reactive library through the use of the `Scheduler` interface. From a practical point of view, this means that it will be possible for different messages, to arrive to an `Observer` from different execution contexts. If all `Observer` instances would have to deal with this scenario, the code in our library would soon become cluttered with concurrency-related housekeeping, making it harder to maintain and reason about.

For this reason, we assume that messages will always arrive in a serialized fashion. As a consequence, operators that deal with events from different execution contexts - e.g. combiner operators - are required to internally perform serialization.

### 3.1.3 Best Effort Cancellation

The next assumption involves premature stream cancellation via `Subscription`s and the function `unsubscribe`, used in order to stop the observation of events from an `Observable`; we are going to assume that whenever `unsubscribe` is invoked, the library will make a best effort attempt to stop all the ongoing work happening in the background. The reason is simple: it is not always safe to abort work that is being processed - e.g. database writes. Although the library might still complete the execution of pending work, its results are guaranteed not to be propagated to any `Observer` that was previously unsubscribed.

### 3.1.4 Resource Cleanup After Termination

As we mentioned in assumption 3.1.1, the guarantee that no events will occur after the first termination message makes it possible to determine when resource cleanup operations are safe to perform. We will now make one step further and assume that resources *will* be cleaned up immediately after termination. This will make sure that any related side-effect will occur in a predictable manner.

### 3.1.5 A note on the Reactive Contract

does the reactive contract even make sense? for example, why do we stop on `onError`? there could be a version of RX which doesn't. Would that make sense? Not sure whether to include this section or not yet.

## 3.2 Concurrency with Schedulers

At the end of Section 2.5 we discussed how `Observable` s, by default, handle data by means of a synchronous pipeline, blocking the processing of successive elements via the call stack. It is worth mentioning again how this synchronous processing does not affect the ability of `Observable` s to handle asynchronous data.

However, this synchronous behavior might not always be the best solution, especially in real world applications, where we might want to have a thread dedicated to listening to incoming events and one which processes them. Enter the `Scheduler` interface, an orthogonal<sup>[20]</sup> structure w.r.t. `Observable` which allows us to introduce concurrency into our reactive equation.

`Scheduler` s allow us to alter the control flow of the data processing within an observable expression, introducing a way to dispatch the work of any number of operators to be executed within the specified context, e.g. a new thread.

The `Scheduler` interface looks like the following<sup>2</sup>.

---

```
31 data Scheduler = Scheduler
32   { _schedule      :: IO () -> IO ()
33     , _scheduleDelay :: IO () -> TimeInterval -> IO ()
34   }
```

---

`Scheduler` s expose two functions which are essentially equal, modulo arbitrary delays in time. Both of these functions take an `IO` action as input and dispatch it to the appropriate execution context, producing a side effect.

To better understand `Scheduler` s, let us present the implementation of one of them, the `newThread` scheduler, which allows us to dispatch actions to a new, dedicated thread.

---

<sup>2</sup>The interface presented in this section is the result of a simplification of the actual one, which involves `Subscription` s. We will discuss the impact of `Subscription` s on `Schedulers` in the next section; suffices to know that the version presented here has no negative effects w.r.t the generality of our discussion.



```
36 newThread :: IO Scheduler
37 newThread = do
38     ch <- newTChanIO
39     tid <- forkIO $ forever $ do
40         join $ atomically $ readTChan ch
41         yield
42     return $ Scheduler (schedule ch) (scheduled ch)
43     where
44         schedule ch io =
45             atomically $ writeTChan ch io
46         scheduled ch io d = do
47             threadDelay $ toMicroSeconds d
48             schedule ch io
```

---

The `newThread` function gives us a side effectful way of creating a `Scheduler` by generating a new execution context - i.e. a new thread - and setting up the necessary tools for safe communication with it. The `Scheduler` functions we are provided, on the other hand, simply write the input `IO` action to the channel and return, effectively dispatching the execution of those actions to the new thread.

Up to this point we haven't mentioned `Observable`s at all. This is the reason why we previously claimed that `Scheduler` and `Observable` are connected by an orthogonal relationship: the two interfaces are independent from one another, yet, when used together within an observable expression, they provide the user with greater expressive power w.r.t. concurrency.

The only thing missing now is a way for us to combine the functionality of these two interfaces: `observeOn` and `subscribeOn` are the operators that will aid us on this task. The former will allow us to dispatch any call to an observer continuation on to the specified execution context, whereas the latter will allow us to control the concurrency of the `Observable` subscribe function.

For the sake of completeness and understandability, the following snippet contains a simple implementation of the `observeOn` operator together with a sample usage.

---

```
50 observeOn :: Observable a -> IO Scheduler -> Observable a
51 observeOn o sched = Observable $ \obr -> do
52     s <- sched
53     subscribe o (f s obr)
```

```
54     where
55         f s downstream = Observer
56             { onNext      = void . _schedule s . onNext downstream
57               , onError    = void . _schedule s . onError downstream
58               , onCompleted = void . _schedule s $ onCompleted downstream
59             }
60
61 obs = Observable $ \obr ->
62     do onNext obr 1
63     onNext obr 2
64     onNext obr 3
65     onCompleted obr
66
67 obr :: Observer Int
68 obr = Observer on oe oc
69     where
70         on v = do
71             tid <- myThreadId
72             print (show tid ++ ": " ++ show v)
73         oe = print . show
74         oc = print "Completed"
75
76 main :: IO ()
77 main = do
78     hSetBuffering stdout LineBuffering
79     subscribe obs' obr
80     tid <- myThreadId
81     putStrLn $ "MainThreadId: " ++ show tid
82     where
83         obs' = obs 'rxmap' (+1) 'observeOn' newThread 'rxmap' (+10)
84         rxmap = flip fmap
85
86 {-
87 output>
88 ThreadId 2: 12
89 ThreadId 2: 13
90 ThreadId 2: 14
91 Completed
```

```
92     MainThreadId: 1
93 -}
```

---

### 3.2.1 A Note on the Concept of Time

Our discussion on push-based collections so far has not once mentioned the concept of time. This might appear strange, especially to the reader familiar with Functional Reactive Programming, where functions over continuous time are at the foundations of the theory. As we mentioned in Section 1.3.3, this dependency on continuous time comes at a great cost: commercial FRP libraries fail to successfully implement the concepts found in the theory as they cannot avoid simulating continuous time and approximating functions operating over it, being this concept inherently discrete in the context of computers.

Rx, on the other hand, completely sheds the notion of time from the notion of reactivity<sup>[15]</sup>, shifting its focus, with the help of `Scheduler` s, to concurrency instead. Time still plays a role, although indirect, within the library: events are processed in the order they happen, and operators make sure such order is maintained, ultimately handing over to the user a stream of time-ordered events.

### 3.2.2 A Note on Orthogonality

Previously we discussed how concurrency is an orthogonal concept w.r.t. Rx - i.e. introducing concurrency does not affect or pollute the definition of our reactive interfaces. This statement is only true from an abstract point of view, falling short of its promises when looking from an implementation perspective, in particular, when dealing with combiner operators (see Section 3.4) such as `(>=)` or `combineLatest` . These operators will not work at their full potential in a synchronous setting, due to the fact that subscribing to a stream will consume it entirely - or forever process, in the case of an infinite stream - before allowing the operator to subscribe to a different one, effectively making interleaving of events impossible.

The problem is gracefully solved with the introduction of `Scheduler` s, which, by allowing for `Observable` s to be executed on different contexts, indirectly make it possible for interleaving to happen and for combiner operators to work at their full potential. This comes at a cost: combiners operators are required to perform message serialization (see assumption 3.1.2) as well as internal state synchronization as, with the introduction of concurrency, messages and state changes can now originate from different execution contexts.

## 3.3 Subscriptions

With schedulers, we are now able to handle observable streams from different execution contexts. The next step in making Rx ready for a production environment is to add a mechanism that will allow

us to stop a stream from anywhere in our program, whenever we don't require its data anymore - i.e. a mechanism that will allow the user to communicate to the `Observable` that one of its `Observer`s is no longer interested in receiving its events.

Is the following clear?

We discussed in the previous section how schedulers effectively boost the expressive power of our reactive expressions by introducing concurrency and interleaving among events originating from different streams. Introducing a cancellation mechanism, on the other hand, is a purely practical concern: although very useful from a practical perspective, especially in the context of resource management, it doesn't impact expressive power from a *reactive* point of view.

The means by which we are going to introduce a cancellation mechanism inside our reactive equation is through the `Subscription` datatype. From a functionality point of view, what we are aiming for is for the `_subscribe` function to hand back a `Subscription` whenever invoked; users will later be able to use this `Subscription` in order to prematurely cease the observation of a stream. This design is closely related to Dispose pattern utilized in the .NET framework<sup>[5]</sup>.

The first step in designing a new feature is to understand how the already existing interfaces will be affected by the newly introduced one; starting from our informal definition of `Observable` from section 2.5, let's now define a new function `(%)`, which incorporates the notion of returning a `Subscription` and see how this is going to affect our types:

---

```
1 %a  :: a -> IO Subscription
2 %?a :: (Either SomeException (Maybe a) -> IO ()) -> IO Subscription
```

---

With this change, each execution of the `Observable` function now returns a `Subscription`, a means for the user to prematurely terminate the processing of the stream.

The next question is the following: to whom does a `Subscription` belong to? The key observation in addressing this question is that an `Observable` can be subscribed to by multiple `Observer`s; our goal is to provide a mechanism that will allow for a fine-grained control over which `Observer` is supposed to stop receiving events. The answer is then straightforward: the notion of subscription is tight to that of observer. The following snippet reflects this observation:

```
1 $a  :: (Subscription, Either SomeException (Maybe a) -> IO ())
2 %$a :: (Subscription, Either SomeException (Maybe a) -> IO ())
3     -> IO Subscription
```

---

Let's quickly summarize what we have discussed so far: a subscription is some object which will allow us to prematurely stop observing a specific stream; since any stream can be subscribed to by multiple observers, we need to associate subscriptions to observers as opposed to observables. Lastly, a subscription is returned every time an observer is subscribed to a stream through the `_subscribe` function. The following modifications to our reactive interfaces reflect these ideas:

---

```
13 newtype Observable a = Observable
14     { _subscribe :: Observer a -> IO Subscription
15     }
16 data Observer a = Observer
17     { onNext      :: a -> IO ()
18     , onError     :: SomeException -> IO ()
19     , onCompleted :: IO ()
20     , subscription :: Subscription
21     }
```

---

So far we have talked a lot about `Subscription`s, yet we haven't clarified what the type really looks like. The general idea is to have `Subscription` record the state of the `Observer` w.r.t. the `Observable` it is subscribed to - be that subscribed or unsubscribed. This can be easily achieved with a variable `_isUnsubscribed :: IORef Bool` initialized to `False`, indicating that the associated `Observer` is initially not unsubscribed.

From a practical point of view, it is useful to augment `Subscription` with some additional functionality. The following code shows a definition of `Subscription` which incorporates an `IO ()` action to be executed at unsubscription time. This is particularly useful when we want to associate resource cleanup actions to the termination - be that forced or natural - of a stream observation. Additionally, it is useful to make the type recursive, allowing `Subscription`s to contain other values of the same type. This will be extremely useful for internal coordination of operators such as `(>=) :: Monad m => m a -> (a -> m b) -> m b`, where each input value will spawn and subscribe a new `Observable`, whose subscription should be linked to the original one. Section 3.4 will extensively discuss this matter.

```
1 data Subscription = Subscription
2   { _isUnsubscribed :: IORef Bool
3     , onUnsubscribe  :: IO ()
4     , subscriptions  :: IORef [Subscription]
5   }
```

---

It's now time to introduce the two functions at the hearth of the whole cancellation mechanism: `unsubscribe` will take care of modifying the state carried by the `Subscription` - i.e. setting `_isUnsubscribed` to `True` - as well as execute the associated `IO ()` action, whereas `subscribe` will simply act as a proxy for the original `_subscribe` function from the `Observable` interface.

---

```
71
72 subscribe :: Observable a -> Observer a -> IO Subscription
73 subscribe obs obr = _subscribe obs safeObserver
74   where
75     safeObserver = Observer safeOn safeOe safeOc s
76     s             = subscription obr
77     safeOn a      = ifSubscribed $ onNext obr a
78     safeOe e      = ifSubscribed $ finally (onError obr e) (unsubscribe s)
79     safeOc        = ifSubscribed $ onCompleted obr >> unsubscribe s
80     ifSubscribed = (>>=) (isUnsubscribed s) . flip unless
81
82 unsubscribe :: Subscription -> IO ()
83 unsubscribe s = do
84   writeIORef (_isUnsubscribed s) True
```

---

The `safeObserver` utilized by the `subscribe` function is of crucial importance to the functionality of our library and its the reason why we need to proxy the original `_subscribe` function: its implementation, in fact, embeds two of the reactive contract assumptions introduced previously. The `safe onNext/onError/onCompleted` functions implement the subscription mechanism, preventing, through the `ifSubscribed` function, events from propagating to the underlying `Observer`, once the related `Subscription` has been unsubscribed. By doing so, it is easy to see how assumption 3.1.3 is satisfied: unsubscribing from a stream does not force the stop of any outstanding work, yet it

is made sure that any result produced after unsubscribing, if any, will not be delivered to the downstream `Observer` - i.e. the `Observer` supplied by the user. Additionally, `safeObserver` allows the enforcement of the reactive grammar seen in assumption 3.1.1; this is done by calling `unsubscribe` as soon as the first termination message - be that `onError` or `onCompleted` - arrives, effectively preventing any additional event from being propagated.

Note that, with the current implementation of the subscription mechanism, an `Observer` can only be subscribed once and only to a single `Observable`, as, once its `Subscription` is unsubscribed, the `_isUnsubscribed` field is never reset to `False`. This convention is shared by many already existing implementations of reactive libraries such as the ones under the ReactiveX umbrella<sup>[21]</sup>.

Now that we have a clear idea of how the subscription mechanism is supposed to work and how it is integrated into our library, let's take a look at a few observations and concerns that involve it.

### 3.3.1 Impact on Schedulers

In the previous sections we discussed a simplified version of the `Scheduler` interface that was glossing, without loss of generality, over details regarding `Subscription`s. In practice, it is useful to associate `Subscription`s not only to `Observer`s but to `Scheduler`s as well.

---

```
95 data Scheduler = Scheduler
96   { _schedule      :: IO () -> IO Subscription
97   , _scheduleDelay :: IO () -> TimeInterval -> IO Subscription
98   , subscription   :: Subscription
99   }
```

---

With this version of the interface, each scheduled action returns a `Subscription`, offering fine grained control over the actions to be executed; at the same time, a `Subscription` is also associated to the `Scheduler` as a whole, allowing the user to perform cleanup actions on the `Scheduler` itself once `unsubscribe` is called. This is best shown with a new example implementation of the `newThread` scheduler and `observeOn` operator:

---

```
101 newThread :: IO Scheduler
102 newThread = do
103   ch <- newTChanIO
104   tid <- forkIO $ forever $ do
```

```
105     join $ atomically $ readTChan ch
106     yield
107     sub <- createSubscription (killThread tid)
108
109     return $ Scheduler (schedule ch) (scheduleD ch) sub
110     where
111         schedule ch io =
112             atomically $ writeTChan ch io
113             emptySubscription
114         scheduleD ch io d = do
115             threadDelay $ toMicroSeconds d
116             schedule_ ch io
117
118 observeOn :: Observable a -> IO Scheduler -> Observable a
119 observeOn o schedIO = Observable $ \obr -> do
120     sched <- schedIO
121     sub <- subscription obr
122     liftIO $ addSubscription sub (subscription sched)
123     _subscribe o (f s obr)
124     where
125         f s downstream = Observer
126             {   onNext      = void . _schedule sched . onNext downstream
127               ,   onError   = void . _schedule sched . onError downstream
128               ,   onCompleted = void . _schedule sched $ onCompleted downstream
129               ,   subscription = subscription downstream
130             }
```

---

The code is mostly equal to the one presented in section 3.2. The most relevant change can be found at line 107, where we create a subscription for the newThread scheduler with an action that simply kills the thread<sup>3</sup>. On line 122 we then add this subscription to the one carried by the downstream observer. In this way, unsubscribing from the downstream subscription will trigger a waterfall effect that will eventually unsubscribing the scheduler's one as well, effectively killing the thread associated to it.

On a last note regarding the relationship between schedulers and subscriptions, it is worth mentioning how the subscription mechanism only works in the presence of schedulers. As we mentioned before, in fact, Rx is synchronous by default in the processing of its data. This means that the program

---

<sup>3</sup>Absolutely not safe, but it's good enough for the sake of our example.



would return from the invocation of the `subscribe` function only after it has fully processed the stream, effectively rendering the subscription mechanism ineffective, as it would not be possible to invoke `unsubscribe` whilst the `Observable` is active. With the introduction of schedules and different execution contexts, this problem disappears and the mechanism works as intended.

### 3.3.2 Formalizing Subscriptions

In Section 2.5 we saw how an `Observable`, at its essence, is nothing more than an instance of the Continuation Monad, where the three types of events that can occur are materialized into a single datatype, `Event a`, as opposed to being handled by three different continuations.

In the discussion that follows we are going to try and understand what the essence of the subscription mechanism is and how it relates to our formal definition of observable as a continuation.

As we mentioned before, a subscription is strongly tight to the notion of observer, as an observable can be subscribed-to multiple times. Although, we can be more specific than this and notice that a subscription is actually tight to the execution of an observable. These two takes on subscriptions are effectively the same thing: an observer can only be subscribed a single time to a single observable, creating the unique link between the subscription and a single execution of the observable function. This perspective is very insightful, as it hints to the fact that a subscription should be immutable within the context of an observable execution. Another observation is that the subscription needs to be retrievable from an observable for a number of reasons, the most important of which being to check whether the subscriber is unsubscribed before pushing any additional events.

The properties of subscription that we just discussed are very similar to the idea of environment variables, shared by computations yet immutable in their nature. The Haskell programming language exposes a monad construct for such computations, the Reader - Environment - Monad: in the remainder of this section, we are going to model the subscription mechanism as a Reader monad transformer on top of our previous definition of observable as a continuation.

---

```
1 type Observer    a = Event a -> IO ()
2 type Observable a = ReaderT Subscription (ContT () IO) (Event a)
3
4 subscribe :: Observable a -> Observer a -> IO Subscription
5 subscribe obs obr = do
6     subscription <- emptySubscription
7     safeObserver  = enforceContract obr
8     runContT (runReaderT obs s) safeObserver
9     return s
```

```
10
11 enforceContract :: Observer a -> Observer a
12 enforceContract obr = ...
```

---

This formalization is very insightful under many points of view: first of all, in the same way as schedulers, it is completely orthogonal to the definition of observable we previously had: the original definition did not change, yet the mechanism was, in a way, glued on top of it. This observation becomes very clear when looking at line 8 in the above snippet: the `subscribe` function first runs the reader transformer, resulting in continuation monad that will have the environment variable available within its context. Notice how each call to `subscribe` will effectively create a new subscription and pair it to the execution of the observable.

A natural question now is: why didn't we use this technique for implementing subscriptions in the "real world" implementation from the previous paragraph? There, we had to change the definitions of our interfaces, losing orthogonality as a consequence. The reason is simply clarity, the interfaces look more clear than reading `readerT`, especially if we want to use that implementation as a reference for other languages. On the other hand, the goal of this paragraph is focusing on the essence of the subscription mechanism, hence the use of theory-related constructs such as monads.

Notice how this formalization only focuses on augmenting the definition of observables with subscriptions. The actual logic of the mechanism remains unchanged and is abstracted away through the `enforceContract` function.

As a final point, it is worth noting that what we presented so far is obviously not the only way we can formalize the subscription mechanism. Many other definitions have been tried out during the course of this work, yet all of the others ended up granting too much or too less power to the resulting mechanism and were therefore discarded. Examples include:

---

```
1 type Observable a = StateT Subscription (ContT () IO) (Event a)
2 type Observable a = Cont () (StateT Subscription IO) (Event a)
```

---

## 3.4 Operators

With schedulers and subscriptions we can handle asynchronous streams of data from different execution context and cease our observation at any point in time. The last feature before we can consider our reactive library ready for use by developers is the introduction of a set of higher order

functions that, given one or more `Observable`s, will allow access to its underlying elements - providing ways to transform, compose and filter them - while abstracting over the enclosing data structure. These higher order functions, also referred to as operators, are a common technique when it comes to abstractions over data structures; examples can be found in many commonly known programming languages, where data structures such as iterables, lists, trees, sets, ..., offer a wide range of functions - `map`, `filter`, `flatMap`, `concat`, etc - that allow the user to operate on its internal elements without requiring any knowledge of the structure that contains them.

### 3.4.1 Functor, Applicative, Monad

Back in section 2.3, we defined the `Functor` instance for the `Observable` type, effectively augmenting our reactive type with our first operator, `map :: (a -> b) -> Observable a -> Observable b`, allowing the user to transform streams of elements of type `a` into streams of elements of type `b` by applying the input function of type `a -> b` to each incoming element in the input stream.

The next operator we are going to introduce is `lift`. This operator takes a function defined on the `Observer` level and lifts it to a more general context, that of `Observable`<sup>[12]</sup>. This pattern is very common in the field of Functional Programming, and it is often used in order to provide an abstraction that facilitates accessing values nested inside container structures such as `Functors`, `Applicatives` or `Monads`. In the context of `Observables`, this function will result very useful as we will be able to define operators on the `Observer` level and later make them accessible in the context of `Observables` simply by lifting the operation.

---

```
23 lift :: (Observer b -> Observer a) -> Observable a -> Observable b
24 lift f ooa = Observable $ \ob -> _subscribe ooa (f ob)
```

---

As an example, we re-propose here the implementation of the `Functor` instance for `Observable`, this time using the `lift` function:

---

```
26 instance Contravariant Observer where
27     contramap f ob =
28         Observer (onNext ob . f) (onError ob) (onCompleted ob) (subscription ob)
29
30 instance Functor Observable where
31     fmap f = lift (contramap f)
```

---

As we discussed in section 2.3, the `Observable` type, at its essence, is an instance of `Applicative Functor` and `Monad` from a category theory perspective. This observation motivates our next effort of trying and defining such instances on the latest version of our interface. We will not bother anymore with proving the associated laws, as it would be a very difficult task, given the complicated nature of the implementation of such functions which now needs to handle subscriptions as well as synchronized state due to schedulers.

Say more about the fact that they don't respect the laws, but just like the `try`, that is not a monad, they are ok? Meaning `try` doesn't respect the laws but nobody really cares, it does what it has to do and from an interface perspective it acts as if it was a monad.

```

139 instance Applicative Observable where
140     pure x = Observable $ \obr -> do
141         onNext obr x
142         onCompleted obr
143         return $ subscription obr
144     (<*>) = combineLatest ($)
145
146 combineLatest :: (a -> b -> r) -> Observable a -> Observable b -> Observable r
147 combineLatest combiner oa ob = Observable $ \downstream ->
148     let
149         onNext_ :: TMVar t -> TMVar s -> (t -> s -> IO ()) -> t -> IO ()
150         onNext_ refT refS onNextFunc valT = join . atomically $ do
151             _ <- tryTakeTMVar refT
152             putTMVar refT valT
153             maybeS <- tryReadTMVar refS
154             return . when (isJust maybeS) $ onNextFunc valT (fromJust maybeS)
155
156         onError_ :: TMVar Bool -> SomeException -> IO ()
157         onError_ hasError e = join . atomically $ do
158             hasE <- takeTMVar hasError
159             putTMVar hasError True
160             return . when (not hasE) $ onError downstream e
161
162         onCompleted_ :: TMVar t -> TMVar Bool -> TMVar Int -> IO ()
163         onCompleted_ refT hasCompleted hasActive = join . atomically $ do
164             emptyT <- isEmptyTMVar refT
165             hasC <- takeTMVar hasCompleted
166             active <- takeTMVar hasActive
167             putTMVar hasCompleted True
168             putTMVar hasActive (active - 1)

```

```
169         return . when (emptyT && not hasC || active - 1 == 0) $
170             onCompleted downstream
171     in do
172         active      <- newTVarIO 2
173         refA        <- newEmptyTVarIO
174         refB        <- newEmptyTVarIO
175         hasError    <- newTVarIO False
176         hasCompleted <- newTVarIO False
177         let obrA = Observer (onNext_ refA refB (fa downstream))
178                         (onError_ hasError)
179                         (onCompleted_ refA hasCompleted active)
180                         (subscription downstream)
181         let obrB = Observer (onNext_ refB refA (fb downstream))
182                         (onError_ hasError)
183                         (onCompleted_ refB hasCompleted active)
184                         (subscription downstream)
185         _subscribe ob obrB
186         _subscribe oa obrA
187     where
188         fa downstream = (\a b -> onNext downstream (combiner a b))
189         fb downstream = (\b a -> onNext downstream (combiner a b))
```

---

The function `pure` does nothing more than wrapping a value into an `Observable` whereas `(<*>)` applies the most recent function emitted by the first `Observable` to the most recent element emitted by the second one. The implementation utilizes the more general `combineLatest` operator, which allows to combine two streams into one by emitting an item whenever either of the two emits one - provided that each of them has emitted at least one.

Last but not least comes the Monad instance for our `Observable` type:

---

```
191 instance Monad Observable where
192     return = pure
193     (>>=)  = flatMap
194
195 flatMap :: Observable a -> (a -> Observable b) -> Observable b
196 flatMap obs f = Observable $ \downstream ->
```

```
197   let
198       onNext_ gate activeRef hasError hasCompleted val = do
199           atomically $ modifyTVar activeRef (+1)
200           s <- emptySubscription
201           let inner = Observer (innerOnNext_)
202                           (onError_ hasError)
203                           (innerOnCompleted_ s)
204                           (s)
205           addSubscription (subscription downstream) (subscription inner)
206           handle (onError_ hasError) . void $ _subscribe (f val) inner
207       where
208           innerOnNext_ v = do
209               withMVar gate $ \_ -> onNext downstream v
210
211           innerOnCompleted_ s = do
212               cond <- atomically $ do
213                   c <- readTVar hasCompleted
214                   modifyTVar activeRef (subtract 1)
215                   a <- readTVar activeRef
216                   return (c && a == 0)
217               if cond
218                   then onCompleted downstream
219                   else removeSubscription (subscription downstream) s
220
221       onError_ hasError e = do
222           cond <- atomically $ do
223               e <- swapTVar hasError True
224               return (not e)
225           when cond $ onError downstream e
226
227       onCompleted_ activeRef hasCompleted = do
228           cond <- atomically $ do
229               c <- swapTVar hasCompleted True
230               a <- readTVar activeRef
231               return (not c && a == 0)
232           when cond $ onCompleted downstream
233
234   in do
```

```

235     gate          <- newMVar ()
236     activeRef     <- newTVarIO (0 :: Int)
237     hasError      <- newTVarIO False
238     hasCompleted  <- newTVarIO False
239
240     _subscribe obs $ Observer (onNext_ gate activeRef hasError hasCompleted)
241                             (onError_ hasError)
242                             (onCompleted_ activeRef hasCompleted)
243                             (subscription downstream)

```

Note how this implementation if (`>=>`) shows the need for the introduction of children subscriptions that we discussed in section 3.3: whenever the outer observable is unsubscribed, we want to automatically unsubscribe any observable that has previously been created by the function passed to (`>=>`).

### 3.4.2 The Essential Operators

Operators can theoretically be infinite in number, as infinite are the transformations that can be done to elements of an observable stream. Practice shows, though, that a relatively small subset of operators and the composition of these, suffices to express the majority of the use cases encountered by users. Leveraging the power of composability of operators is advantageous towards the design of a simple yet powerful API.

Operators can be grouped into categories by looking at their characteristics; it is not the purpose of this work to list every possible operator and its semantics, yet, for the sake of completeness, table 3.1 presents the most important categories<sup>4</sup> and the associated operators:

Put operator  
description in  
the table?

### 3.4.3 Formalizing Operators

So far we have seen which are the most useful operators and how they aid in making our reactive library more useful from a user perspective. Things get more complicated when we try to analyze them from a theoretical point of view: operators can be viewed as state machines, containing an internal state which is modified whenever an event occurs, following the semantics of the specific operator.

For this reason, trying to formalize them as a functional and pure structure becomes very difficult, resulting in more confusion than clarity. This outcome should not come as a surprise: operators

<sup>4</sup>An implementation of these operators can be found at the repository associated with this work.

Table 3.1: The essential operators

<b>Transformation</b>	buffer bufferWithSkip groupBy fmap scan scanLeft sample throttle window
<b>Filtering &amp; Conditional</b>	filter distinctUntilChanged skip skipUntil skipWhile take takeUntil untilTake
<b>Combining</b>	(»=) (<*>) concat startsWith withLatestFrom zip zipWithBuffer
<b>Error Handling</b>	catchOnError onErrorResumeNext retry
<b>Utility</b>	observeOn subscribeOn publish share ofType doOnNext doOnError doOnCompleted toIterable toList



defined on other more common data structures such as lists or trees are state machines as well, usually hiding their state using function parameters:

---

```
1 take :: Int -> [a] -> [a]
2 take _ []      = []
3 take 0 _       = []
4 take n (x:xs) = x : take (n-1) xs
```

---

As we can see from this example implementation of take on lists, the internal state of the operator - i.e. the number of elements to be taken, `n` - is wired in the definition of the function, eliminating the need for an internal variable as a result.

This type of operator definition works very neatly for any pull-based data structure, where we can define the operators recursively on the structure of the collection. For `Observable` things become a little more complicated since we are dealing with a push-based collection, where elements are never gathered as a concrete collection in memory, not allowing, as a consequence, any type of structural recursion.

We will discuss in section 3.4.3, Future Work, how Event Calculus might be used as a technique to better define the semantics of operators for push-based collections.



## CONCLUSION

*TBD*

---

— Lewis Carol,  
*Alice in Wonderland*

Bla bla bla

**Future Work**

**Conclusion**



## TODO LIST

■ Finish it! . . . . .	i
■ Fix repo organization . . . . .	3
■ Talk about FRP, Reactive Streams, Rx and other stuff. Cite survey paper Claim that Observable is dual of Iterator in order to justify the why. . . . .	7
■ Probably put the derivation of Reactive Streams in the appendix, it's not really the focus, I guess. . . . .	7
■ Say something about why I chose to go with the RxJava style of implementation? . . . . .	29
■ does the reactive contract even make sense? for example, why do we stop on onError? there could be a version of RX which doesn't. Would that make sense? Not sure whether to include this section or not yet. . . . .	32
■ Is the following clear? . . . . .	36
■ Say more about the fact that they don't respect the laws, but just like the try, that is not a monad, they are ok? Meaning try doesn't respect the laws but nobody really cares, it does what it has to do and from an interface perspective it acts as if it was a monad. . . . .	44
■ Put operator description in the table? . . . . .	47





**B**egins an appendix

**A.1 Categorical Duality**

**A.2 Continuations**

**A.3 (Co)Products**

**A.4 (Un)Currying**

**A.5 (Covariance & Contravariance**

**A.6 Functors**







## APPENDIX B

**B**egins an appendix  
Formal RX Implementation



## BIBLIOGRAPHY

- [1] *Scopus search: Reactive programming 1960-2016.*
- [2] *Rx Design Guidelines.*  
<http://go.microsoft.com/fwlink/?LinkID=205219>, October 2010.  
Accessed: 2015-12-28.
- [3] E. BAINOMUGISHA, A. L. CARRETON, T. V. CUTSEM, S. MOSTINCKX, AND W. D. MEUTER, *A survey on reactive programming*, ACM Computing Surveys (CSUR), 45 (2013), p. 52.
- [4] A. BENVENISTE AND G. BERRY, *The synchronous approach to reactive and real-time systems*, Proceedings of the IEEE, 79 (1991), pp. 1270–1282.
- [5] K. CWALINA AND B. ABRAMS, *Framework design guidelines: conventions, idioms, and patterns for reusable. net libraries*, Pearson Education, 2008.
- [6] J. EDWARDS, *Coherent reaction*, in Proceedings of the 24th ACM SIGPLAN conference companion on Object oriented programming systems languages and applications, ACM, 2009, pp. 925–932.
- [7] D. FAHLAND, D. LÜBKE, J. MENDLING, H. REIJERS, B. WEBER, M. WEIDLICH, AND S. ZUGAL, *Declarative versus imperative process modeling languages: The issue of understandability*, in Enterprise, Business-Process and Information Systems Modeling, Springer, 2009, pp. 353–366.
- [8] B. FURHT AND A. ESCALANTE, *Handbook of cloud computing*, vol. 3, Springer, 2010.
- [9] E. GAMMA, *Design patterns: elements of reusable object-oriented software*, Pearson Education India, 1995.
- [10] T. GOWERS, J. BARROW-GREEN, AND I. LEADER, *The Princeton companion to mathematics*, Princeton University Press, 2010.
- [11] C. H. GROUP, *Covariance, contravariance, and positive and negative position.*
- [12] HASKELL.ORG, *Lifting.*

## BIBLIOGRAPHY

---

- [13] R. K. JONAS BONER, DAVE FARLEY AND M. THOMPSON, *"reactive manifesto"*, "2014".
- [14] D. LANEY, *3d data management: Controlling data volume, velocity and variety*, META Group Research Note, 6 (2001), p. 70.
- [15] E. MEIJER, *Subject/Observer is Dual to Iterator*, in FIT: Fun Ideas and Thoughts at the Conference on Programming Language Design and Implementation, 2010.
- [16] ———, *Your mouse is a database*, Queue, 10 (2012), p. 20.
- [17] ———, *Reactive streams keynote*.  
LambdaJam, 2014.
- [18] E. MEIJER, K. MILLIKIN, AND G. BRACHA, *Spicing up dart with side effects*, Queue, 13 (2015), p. 40.
- [19] J. NEWBERN, *All about monads*.
- [20] ORTHOGONALITY, *"orthogonality — Wikipedia, the free encyclopedia"*, "2016".
- [21] REACTIVEX, *"rxjava - wiki"*, "2015".
- [22] TEL, *What's up with contravariant?*